

Numerical Integration and Differentiation Example

A manufacturer has the capability of transforming thin, flat sheets of metal into uniformly corrugated sheets of various pre-determined thicknesses.

The corrugation machine transforms flat sheets into rippled sheets of the form

$$y = 0.5 * t * \sin(\pi * x)$$


where y is the displacement of the sheet from the original horizontal plane x inches from the initial edge of the corrugation. The parameter t is the effective thickness of corrugated sheet. (To simplify this example, we assume that the material thickness is negligible.)

The plant is only able to cut flat sheets. Corrugated sheets cannot be cut, so the sheets must be shortened prior to corrugation.

Suppose the manufacturer currently has flat sheets in stock which are 60 cm long. What is the resulting length of the corrugated sheets if the thickness is set to 1 cm?

We can use the fact that the length of a curve is defined by the equation

$$\text{length} = \int [1 + (dy/dx)^2]^{0.5}$$



where length is the length of the flat sheet and dy/dx is the derivative of y with respect to horizontal location, x . If the integration is done from 0 to the length of the corrugated sheet, the result will be the length of the original flat sheet.

Simpson's method for numerical integration involves the generation of a number of function slices, each of which has an area. The resulting integral is computed as the sum of the areas for all the slices.

Here is a TK Solver procedure which computes the integral using Simpson's method. Each step in the process is commented within the procedure.

```

-----
; t is the thickness of the corrugated sheet
; dx is the differentiation increment; relatively small compared to typical x values
; a is the lower limit of integration
; b is the upper limit of integration
; n is the number of integration slices

; Integration by Simpson's method
h = (b-a)/n ; width of each slice
x = a ; start at the lower limit
k = 1 ; Simpson's k
length = 0 ; initialize the resulting sum variable

for i=2 to n ; loop through the slices
  x = x + h ; increment to the next width
  k = 3 - k ; update k

  ; Differentiate the corrugate function at the current value of x
  y1 = t/2*sin(pi()*(x+dx)) ; evaluate the corrugate function at x+dx
  y2 = t/2*sin(pi()*(x-dx)) ; evaluate the corrugate function at x-dx
  z = (y1-y2)/(2*dx) ; compute the slope (dy/dx)

  ; Define the integrand
  int = sqrt(1+z^2)
  length = length + k*int ; compute the area of the current slice
next i

; Account for the areas at the ends
ya1 = t/2*sin(pi()*(a+dx))
ya2 = t/2*sin(pi()*(a-dx))
za = (ya1-ya2)/(2*dx)
inta = sqrt(1+za^2)

yb1 = t/2*sin(pi()*(b+dx))
yb2 = t/2*sin(pi()*(b-dx))
zb = (yb1-yb2)/(2*dx)
intb = sqrt(1+zb^2)

; Total length
length = (2*length + inta + intb)*h/3

```

This procedure requires five inputs – t, dx, a, b, n – and returns the resulting length.

A single rule is used to reference the procedure from the TK Solver Rule Sheet.

```
length = allin1(t,dx,LL,UL,slices)
```

The variables are summarized on the TK Solver Variable Sheet.

St	Input	Name	Output	Unit	Comment
	1	t			Thickness of corrugated sheet
	0	LL			Lower limit of integration -- start of corrugated sheet (0)
	50	UL			Upper limit of integration -- length of corrugated sheet
	1000	slices			Integration slices
	.001	dx			Differentiation increment
		length	73.1847147		Length of flat sheet

We see that if the corrugated sheet is 50 cm, the flat sheet needs to be 73.18 cm. We can use TK Solver's Iterative Solver to backsolve the problem to determine the corrugated sheet length if the flat sheet is initially 60 cm. Input 60 for length and change the status of UL to Guess and solve.

St	Input	Name	Output	Unit	Comment
	1	t			Thickness of corrugated sheet
	0	LL			Lower limit of integration -- start of corrugated sheet (0)
		UL	40.9938433		Upper limit of integration -- length of corrugated sheet
	1000	slices			Integration slices
	.001	dx			Differentiation increment
	60	length			Length of flat sheet

The solution is 40.99 cm.

The TK Solver procedure solves this specific problem but it is possible to use object-oriented programming in TK Solver to make the solution process generic. For example, consider what changes would be required if the corrugation function changed. Six lines of code refer to the corrugation equation and would all need to be edited. It would be much more efficient if the corrugation equation could be defined as a function and then repeatedly accessed wherever it is required. The same logic applies to the differentiation and integration functions used in this example.

Step 1: Define the corrugation function.

Go to the Function Sheet and create a Procedure Function called **corrugate**. Open the subsheet and enter the equation along with the variable definitions.

Comment: form of the corrugated sheet

Parameter Variables: **t**

Input Variables: **x**


Output Variables: **y**

$$y = (t/2) * \sin(\pi() * x)$$

The input to the function, x , will be varied by the differentiation routine to evaluate the result, y . The parameter variable, t , remains constant throughout the process and is pulled directly from the variable sheet. Close the function subsheet and go to the Rule Sheet to test it. Enter the rule

$$y = \text{corrugate}(x)$$

Switch to the Variable Sheet and enter inputs of **1** and **.2** for t and x , then solve. $y = 0.29389263$. This means that 0.2 cm from the end of the corrugated sheet, the sheet is displaced almost 0.3 cm from the horizontal plane. Here is a TK plot of the function from $x = 0$ to $x = 5$.



Step 2: Define the integrand of the length equation.

$$[1 + (dy/dx)^2]^{0.5}$$

This expression can be entered as another procedure function. In the above equation we need to evaluate the derivative (dy/dx). We can create a procedure function to do the differentiation using a simple slope formula.

Create a procedure function called `diff13` on the function sheet. Open the subsheet for `diff13`. The procedure will require three inputs – the name of the function to be differentiated, the point of differentiation, and a relatively small increment over which the slope is computed. You can use TK's built-in `APPLY` function to evaluate the function to be differentiated at $x+h$ and $x-h$ as shown below. A single line of code does the job.

Comment: First derivative, 3-point formula

Parameter Variables:

Input Variables: `fun,x,h`

Output Variables: `z`

; Notation: `fun` - name of the function to differentiate

; `x` - point of differentiation

; `h` - increment

`z = (apply(fun,x+h)-apply(fun,x-h))/(2*h)`

Close the subsheet for `diff13`.

Step 4: Define the integrand function.

Create a procedure function called **integrand**. Open the subsheet and enter the equation and variable definitions.

Comment: function to be integrated

Parameter Variables:

Input Variables: `c`

Output Variables: `f`

`f = sqrt(1 + (diff13('corrugate,c,.001))^2)`

The variables `c` and `f` represent the lengths of the corrugated and flat sheets. Close the integrand subsheet.

Step 5: Create a numerical integration tool based on Simpson's method.

Create a procedure function called Simpson and enter the following statements.

```
-----  
; fun   name of rule, procedure, list or built-in function  
;       defining the integrand  
; a,b   lower and upper limits  
; n     number of integration steps (must be an even number)  
; value value of definite integral  
  
; Description: Standard method of numerical integration using second-degree  
;              polynomial approximation of integrand.  
  
h= (b-a)/n  
x= a  
k= 1  
value= 0  
for i=2 to n  
  x= x + h  
  k= 3 - k  
  value= value + k*apply(fun,x)  
next i  
value= (2*value+apply(fun,a)+apply(fun,b))*h/3  
-----
```

Simpson requires 4 inputs -- the name of the integrand function, the integration limits, and the number of integration slices. Close the subsheet and switch to the Rule Sheet. Use a rule such as

$$LF = \text{Simpson}(\text{'integrand}, 0, LC, 1000)$$

where LF and LC are the lengths of the flat and corrugated sheets. This rule uses 1000 slices to do the computation. Enter the following rule to compute the integral.

Open the Variable Sheet, enter **60** as input for **LF** and guess **LC** as **30**. Solve, and **LC = 41**. Do you know why a guess was necessary for LC? Without it, the Simpson procedure would not have values for all of its inputs.

This process provides the same solution as the single procedure but it is easier to adapt to new problems. Suppose the manufacturer has a second corrugation machine that produces sheets of the form

$$y = (t/3) * \sin(\pi() * x)$$

Just edit the corrugate function and you're ready to go.